

水の熱力学的性質を計算するプログラム — Haar et al. (1984) の式を用いて —

A computer code for thermodynamic properties of water with the use of the equation of Haar et al. (1984)

澁江 靖 弘*
SHIBUE Yasuhiro

This paper shows a BASIC program for the computation of thermodynamic properties of water. The program is developed from the FORTRAN code described by Haar et al. (1984). This program runs on Windows® machine (either Windows® XP or Windows® 2000).

キーワード：コンピュータプログラム, BASIC, 水, 熱力学的性質

Key words : Computer program, BASIC, Water, Thermodynamic properties

1. はじめに

水の熱力学的性質（密度, エンタルピー, ギブスエネルギー, 比熱など）は, 地熱熱水系のモデリングを行う上で必要不可欠である。また, 塩化ナトリウムなどの電解質が溶解している水溶液の熱力学的性質を求める上でも必要である。水の熱力学的性質に関する研究は多数行われており, それらの研究をまとめた蒸気圧表 (Steam Tables) も多くの研究者達によって作成されている。本研究のベースになった Haar et al. (1984) の “Steam Tables” もその内の一つである。現在, 最も正確に水の熱力学的性質を表す式は Wagner and Pruß (2002) であるが, 370°C 付近を除いて Haar et al. (1984) の “Steam Tables” の誤差は極めて小さい (例えば, 澁江, 2003)。また, 電解質水溶液の熱力学的性質が Haar et al. (1984) の式に基づいていることも多く (例えば, 澁江, 1997, 2003), 今日においても Haar et al. (1984) の式の利用価値は高い。筆者は彼らが示した FORTRAN のプログラムを参考にして Windows® (Windows® 2000 と Windows® XP) 上を走る BASIC のプログラムを作成した (この時のシステムとして 電脳組の BASIC/98® を使用している)。飽和蒸気圧条件下での計算に必要なプログラムは先に示した (澁江, 2005) が, 本論文では一般的な計算ができるプログラムを示す。得られる量は, 圧力 (密度と温度を指定した場合) あるいは密度 (温度と圧力を指定した場合), 密度一定の条件下での圧力の温度微分, 温度一定の条件下での圧力の密度微分, 定圧比熱, 定積比熱, エントロピー, エンタルピー, 内部エネルギー, ギブスエネルギーである。先に報告した気液二相共存条件下で

のプログラムは圧力条件を限定しているために計算結果が速く得られると言う利点を持つ。計算する条件を考慮して, 本プログラムと使い分けると良からう。

2. Haar et al. (1984) の状態方程式

Haar et al. (1984) は, ヘルムホルツエネルギー (A) を密度 (ρ) と絶対温度 (T) の関数 (ヘルムホルツ関数) として表した。ヘルムホルツ関数は, Base function (A_{base}), Residual function (A_{residual}), Ideal gas function ($A_{\text{ideal gas}}$) の3つの部分からなる。これらの関数 (function) とその係数については先の報告で示したので, ここでは, 繰り返しを避ける。本プログラムで計算する性質の間の関係を表1にまとめて示す。

Haar et al. (1984) の式を適用できる温度と圧力の領域は次の通りである。温度は 273.15K 以上 1273.15K 以下である。温度が 423.15K 以上の時には, 適用可能圧力の最大値は 15kbars である。温度 (T) が 273.15K 以上 423.15K 未満の時では, 適用可能圧力の最大値 (P_{max}) は次式で計算できる。

$$P_{\text{max}} \text{ (kbars)} = 5 + (T - 273.15) / 15$$

純水の臨界点付近ではこの式はあまり正確ではない。Haar et al. (1984) は臨界点での温度, 圧力, 密度を 647.126K, 220.55bars, 0.322 (g/cm³) と置いた。そして, 温度 (T, 単位はK) あるいは密度 (ρ , 単位はg/cm³) が 647.026 < T < 647.226 あるいは, 0.322 × 0.7 < ρ < 0.322 × 1.3 の領域に入っている時にはその他の領域に比べて

*兵庫教育大学第3部 (自然系教育講座)

表1 熱力学的性質（ヘルムホルツエネルギー、エントロピー、内部エネルギー、エンタルピー、ギブスエネルギー、定積比熱、定圧比熱）と圧力や密度や温度との関係。

ヘルムホルツエネルギー(A)と圧力(P), 密度(ρ), 温度(T)との関係

$$P = \rho^2 \left(\frac{\partial A}{\partial \rho} \right)_T$$

$$\left(\frac{\partial P}{\partial \rho} \right)_T = \frac{2P}{\rho} + \rho^2 \left(\frac{\partial^2 A}{\partial \rho^2} \right)_T$$

$$\left(\frac{\partial P}{\partial T} \right)_\rho = \rho^2 \left(\frac{\partial^2 A}{\partial \rho \partial T} \right)_\rho$$

ヘルムホルツエネルギー(A)とエントロピー(s)

$$s = - \left(\frac{\partial A}{\partial T} \right)_\rho$$

ヘルムホルツエネルギー(A)と内部エネルギー(u)

$$u = A + Ts$$

ヘルムホルツエネルギー(A)とエンタルピー(h)

$$h = u + \frac{P}{\rho}$$

ヘルムホルツエネルギー(A)とギブスエネルギー(G)

$$G = A + \frac{P}{\rho}$$

ヘルムホルツエネルギー(A)と定積比熱(c_v)

$$c_v = -T \left(\frac{\partial^2 A}{\partial T^2} \right)_\rho$$

ヘルムホルツエネルギー(A)と定圧比熱(c_p)

$$c_p = c_v + \left(\frac{T}{\rho^2} \right) \left(\frac{\partial P}{\partial T} \right)_\rho^2 \left(\frac{\partial \rho}{\partial P} \right)_T$$

不正確であるとしている。飽和蒸気圧条件では、温度が646.3Kより高く647.126K以下の時には次の近似式で液相の密度(ρ_l)と気相の密度(ρ_v)が計算されている。

$$\rho_l = 0.322 + 0.657128 (1 - T/647.126)^{0.325} \quad (1)$$

$$\rho_v = 0.322 - 0.657128 (1 - T/647.126)^{0.325} \quad (2)$$

そして式(2)から計算できる気相の密度と入力した温

度から飽和蒸気圧を計算している(つまり、液相の密度は不正確であるために飽和蒸気圧の計算に用いられていない)。飽和蒸気圧以外の熱力学的性質の計算は温度と式(1)と式(2)で計算する密度の値を用いて行う。気液二相共存条件でこの密度に関する近似式を用いると、気相と液相のギブスエネルギーの計算値の違いが0.015(J/g)程度になる。先に報告したプログラムでは647.073K以上で臨界温度までの飽和蒸気圧の計算ができないことになっているが、上の二つの式を用いれば計算可能で

ある。この場合、澁江 (2005) 中のList 1として示されているプログラムのline 26800中のTZ=647.073をTZ=647.126に改める必要がある。ただし、最近の研究によれば、水の臨界温度は647.096K (Wagner and Pruß, 2002) であるのでこの温度以上での飽和蒸気圧の値は無意味である。

3. コンピュータプログラム

著者が作成したプログラムを本論文末尾にList 1として示す。プログラムリスト中には変数の型指定が抜けているものがあるが、コンピュータ内部では倍精度変数として取り扱われている。List 1中のサブルーティンは飽和蒸気圧の計算に用いたサブルーティン (澁江, 2005) と同じものである。ただし、いくつかのサブルーティンには多少の変更を加えている。変更点は余計な演算を省略したことと変数名を変えたこと (表2) であるので、サブルーティンの内容を多少変更したList 1のプログラ

ムで飽和蒸気圧条件下での密度や熱力学的性質を計算しても先に示した飽和蒸気圧条件下での計算プログラム (澁江, 2005) と同じ結果になる。変更点のないサブルーティンについては、List 1中で行番号に続いて「サブルーティン (*□□□…)」と記した。括弧内の□□□…の部分は該当するサブルーティン名である。先に示した報告中のサブルーティン中のGOTO文の行番号だけを適切に変換すればそのまま用いることができる。GOTO文はすべて同一のサブルーティン内への移動を示すだけであるので、変換は容易である。

このプログラム全体に対する説明は紙数の関係で不可能である。そこで、計算方法を (1) 温度と密度を入力して圧力を計算する場合と (2) 温度と圧力を入力して密度を計算する場合に分けて、以下に簡単に示す。計算で使用するサブルーティンを名前の前に*を付けて示す。

表2 澁江 (2005) 中で示したサブルーティンからの変更点。

サブルーティン	行番号 (List 1)	変更点
*CORRTPDLVDVDELG	17400	DP=DPD を取り除いた。
*CORRTPDLVDVDELG	17650	DVV=DVAP を挿入した。
*CORRTPDLVDVDELG	18100	STOP を RETURN に改めた。
*CORRTPDLVDVDELG	18200, 18250 , 18300	TAU を TAUC に改めた。
*CORRTPDLVDVDELG	18550	P を PPP に改めた。
*BASEDT	20100	-LOG (XX) を (-1#) * LOG (XX) に改めた。
*BASEDT	20300	-T を (-1) * T に改めた。
*QQTD	21000	-A を (-1) * AA に改めた。
*QQTD	21650	QK=CDBL (K) : QKM=CDBL (KM) を挿入した。
*QQTD	22950	-AAD (J-36) を (-1#) * AAD に改め、CDBL (K) を QK に改めた。
*QQTD	22950	CDBL (KM) を QKM に改めた。
*QQTD	23150	-ATT を (-1#) * ATT に改めた。
*QQTD	23300, 23400	CDBK (K) を QK に改め、CDBL (KM) を QKM に改め、CDBL ((K-1)) を (QK-1#) に改め、(K-2) を (QK-2#) に改め、2 となっていた所を 2# に改めた。
*QQTD	23800	-DADT を (-1#) * DADT に改めた。
*IDEALT	25750	右辺の負号を (-1#) に改めた。
*IDEALT	26000, 26050	(I-6) を CDBL (I-6) に改めた。
*PCORRTPDLVDV	26500	P=PS:PPP=P を PPP=PS に改めた。
*PCORRTPDLVDV	26600	DP=0 を挿入した。
*PCORRTPDLVDV	26700	P=P+DP を PPP=PPP+DP に改めた。
*PCORRTPDLVDV	26800	PPP=P を取り除いた。
*PCORRTPDLVDV	26850	P=PPP を挿入した。

3.1 温度と密度を入力して圧力を計算する場合

圧力の計算は次のようにして行う。入力した温度を絶対温度で表し (*TTTT), *BBTと*BASEDTを用いて $(\partial A_{base}/\partial \rho)_T$ を計算する。*QQT Dにおいて $(\partial A_{residual}/\partial \rho)_T$ を計算する。 $A_{ideal\ gas}$ は密度に依存しないので、 $(\partial A/\partial \rho)_T$ は $(\partial A_{base}/\partial \rho)_T$ と $(\partial A_{residual}/\partial \rho)_T$ の和に等しい。圧力は $\rho^2 (\partial A/\partial \rho)_T$ に等しいので、 $(\partial A_{base}/\partial \rho)_T + (\partial A_{residual}/\partial \rho)_T$ に密度の二乗を掛け合わせた値になる。実際には、 A_{base} や $A_{residual}$ は無次元化されているので、計算では適切な定数項が掛けられている。その後で、*THERMDTのサブルーティンでエンタルピーなどの熱力学的性質を計算する。

プログラムの入力例と出力例をそれぞれ表3と表4に示す。入力例を示す表中にはプログラムからの問いかけについての説明を加えている。また、出力例を示す表には、出力値についての説明を加えている。

3.2 温度と圧力を入力して密度を計算する場合

温度が臨界温度よりも高い場合には、温度と圧力から1通りの密度が計算できるはずである。温度が臨界温度よりも低い場合には、圧力が飽和蒸気圧に等しい場合と

等しくない場合に分ける必要がある。圧力が飽和蒸気圧に等しくない時にも密度の値は1通りに決まる。飽和蒸気圧に等しい時には、密度の値が2つ得られる(液相の密度と気相の密度)。そこで、温度の入力値が臨界温度より低温の場合には飽和蒸気圧をまず計算する。この計算は*PCORRTPDL DVのサブルーティンで澁江(2005)の図1中で示されたフローシートに沿って行う。なお、この計算では、液相のギブスエネルギーから気相のギブスエネルギーを引いた値を気体定数と絶対温度の積で割った計算結果(List 1中の行番号18550で計算するDELG)の絶対値が0.0001未満になった時に気液二相が平衡状態にあるとしている。Haar et al. (1984)の式は臨界点付近で不正確であるので、プログラムでは仮想的な臨界温度として647.073Kの値が使用されている。また、温度の入力値から計算できる飽和蒸気圧($P_{vap-sat}$)と圧力(P)が、 $-0.00005 \leq (P - P_{vap-sat})/P_{vap-sat} \leq 0.00005$ の関係式を満たす場合には、圧力を飽和蒸気圧とみなして液相と気相の性質を計算する。

温度が647.073Kよりも高温の場合と647.073Kよりは低温であるが飽和蒸気圧よりも高圧である場合には、密度の初期推定値を $2.5P/T$ と置く。温度が647.073Kよりも低

表3 プログラムの入力例。密度と温度を入力して圧力を計算する場合。入力値について矢印をつけて説明を加えている。

```

Run
*****
* Enter units      *
*****
TEMPERATURE
Choose from 1=deg K, 2=deg C←温度の単位を選ぶ
? 2←2の摂氏温度を選んだ
DENSITY
Choose from 1=kg/m3, 2=g/cm3←密度の単位を選ぶ
? 2←2のg/cm³を選んだ
PRESSURE
Choose from 1=MPa, 2=bar, 3=atm, 4=PSIA, 5=kg/cm2←圧力の単位を選ぶ
? 2←2のbarを選んだ
ENERGY
Choose from 1=kJ/kg, 2=J/g, 3=J/mol, 4=cal/g, 5=cal/mol←エネルギーの単位を選ぶ
? 2←2のJ/gを選んだ

Enter option and temperature. Option=1, input density. Option=2, input pressure.
Option No.? 1←密度の入力を選んだ
Density or Pressure? 0.75←密度の値を0.75(g/cm³)と入力した
Temperature? 300←300℃と入力した
Will you continue the calculation? Input Y(or y) or N(or n)? n
OK←計算が終了するとこの上の行の問いが出る。計算を続けるときにはYあるいはyを入力し、終了する時にはNあるいはnを入力する。この場合は終了することを選んだのでOKのメッセージが出てプログラムは終了した。
    
```

表4 計算結果の出力例。出力値について矢印をつけて説明を加えている。

```

Units←入力した単位一覧
TEMPERATURE  C
DENSITY  g/cm3
PRESSURE  bar
ENERGY  J/g

T= 300.0000    P= 293.671003    D= 0.750000000000
DP/DT=+0.11738685E+02  DP/DD=+0.676748552E+04
CP=+0.508798015E+01    CV=+0.30132704E+01
S= 3.176292    H= 1328.260813    U= 1289.104679
G= -492.230761    A= -531.386894
    
```

出力中の各記号は以下の値を示す。
 T：入力した温度 (°C), P：飽和蒸気圧 (bar), D：密度 (g/cm³),
 DP/DT：密度一定の条件での圧力の温度微分 (bar/K),
 DP/DD：温度一定の条件下での圧力の密度微分 (bar•cm³/g),
 CP：定圧比熱 (J/g•K), CV：定積比熱 (J/g•K),
 S：エントロピー (J/g•K), H：エンタルピー (J/g), U：内部エネルギー (J/g),
 G：ギブスエネルギー (J/g), A：ヘルムホルツエネルギー (J/g)

表5 プログラムの入力例。圧力と温度を入力して密度を計算する場合。入力値について矢印をつけて説明を加えている。

```

Run
*****
* Enter units      *
*****
TEMPERATURE
Choose from 1=deg K, 2=deg C←温度の単位を選ぶ
? 2←2 の摂氏温度を選んだ
DENSITY
Choose from 1=kg/m3, 2=g/cm3←密度の単位を選ぶ
? 2←2 の g/cm3を選んだ
PRESSURE
Choose from 1=MPa, 2=bar, 3=atm, 4=PSIA, 5=kg/cm2←圧力の単位を選ぶ
? 2←2 の bar を選んだ
ENERGY
Choose from 1=kJ/kg, 2=J/g, 3=J/mol, 4=cal/g, 5=cal/mol←エネルギーの単位を選ぶ
? 2←2 の J/g を選んだ

Enter option and temperature.  Option=1, input density.  Option=2, input pressure.
Option No.? 2←圧力の入力を選んだ
Density or Pressure? 1000←圧力の値を 1000bars と入力した
Temperature? 500←500°Cと入力した
Will you continue the calculation?  Input Y(or y) or N(or n)? n
OK←計算が終了するとこの上の行の問いが出る。計算を続けるときにはYあるいはyを入力し、終了する時にはNあるいはnを入力する。この場合は終了することを選んだのでOKのメッセージが出てプログラムは終了した。
    
```

表6 計算結果の出力例。出力値について矢印をつけて説明を加えている。

```

Units←入力した単位一覧
TEMPERATURE  C
DENSITY  g/cm3
PRESSURE  bar
ENERGY  J/g

T=  500.0000      P=  1000.000000  D=  0.528211380566
DP/DT=+0.60517930E+01  DP/DD=+0.348856629E+04
CP=+0.555736012E+01    CV=+0.26481881E+01
S=  4.489707      H=  2316.229382    U=  2126.911239
G= -1154.987592    A= -1344.305735
    
```

出力中の各記号は以下の値を示す。

T : 入力した温度 (°C), P : 飽和蒸気圧 (bar), D : 密度 (g/cm³),
 DP/DT : 密度一定の条件での圧力の温度微分 (bar/K),
 DP/DD : 温度一定の条件下での圧力の密度微分 (bar•cm³/g),
 CP : 定圧比熱 (J/g•K), CV : 定積比熱 (J/g•K),
 S : エントロピー (J/g•K), H : エンタルピー (J/g), U : 内部エネルギー (J/g),
 G : ギブスエネルギー (J/g), A : ヘルムホルツエネルギー (J/g)

表7 プログラムの入力例。気液二相平衡条件かそれに極めて近い圧力と温度を入力して密度を計算する場合。入力値について矢印をつけて説明を加えている。

```

Run
*****
* Enter units      *
*****
TEMPERATURE
Choose from 1=deg K, 2=deg C←温度の単位を選ぶ
? 2←2 の摂氏温度を選んだ
DENSITY
Choose from 1=kg/m3, 2=g/cm3←密度の単位を選ぶ
? 2←2 の g/cm3 を選んだ
PRESSURE
Choose from 1=MPa, 2=bar, 3=atm, 4=PSIA, 5=kg/cm2←圧力の単位を選ぶ
? 2←2 の bar を選んだ
ENERGY
Choose from 1=kJ/kg, 2=J/g, 3=J/mol, 4=cal/g, 5=cal/mol←エネルギーの単位を選ぶ
? 2←2 の J/g を選んだ

Enter option and temperature.  Option=1, input density.  Option=2, input pressure.
Option No.? 2←圧力の入力を選んだ
Density or Pressure? 85.837843←圧力の値を 85.837843bars と入力した
Temperature? 300←300°C と入力した
Will you continue the calculation?  Input Y(or y) or N(or n)? n
OK←計算が終了するとこの上の行の問いが出る。計算を続けるときには Y あるいは y を入力し、終了する時には N あるいは n を入力する。この場合は終了することを選んだので OK のメッセージが出てプログラムは終了した。
    
```

く圧力が飽和蒸気圧よりも低い場合には、密度の初期推定値を 0 と置く。これらの初期推定値を用いて $P - \rho^2 (\partial A / \partial \rho)_T = 0$ の関係式が成立するような ρ の値を逐次近似を繰り返して求める。この計算はサブルーティン*DFIN DDOUTTPDTPDPを用いて行う。このようにして求められた密度の値を用いて、サブルーティン*THERMDTより熱力学的性質を計算する。

プログラムの入力例と出力例をそれぞれ表 5 から表 8 に示す。入力例を示す表中にはプログラムからの問いかけについての説明を加えている。また、出力例を示す表には、出力値についての説明を加えている。表 5 は臨界温度より高い温度を入力した時の例であり、表 6 はこの時の出力である。表 7 は飽和蒸気圧を入力した時の例であり、表 8 はこの時の出力である。表 8 中の G の値（ギブスエネルギーの値）を液相と気相の間で比較すると、若干違っている。これは、入力した圧力が厳密には飽和蒸気圧と一致しないことと、このプログラムでは、液相と気相の 1 g あたりのギブスエネルギーの差が $\pm 0.0001\text{J}$ 以内であれば気液二相が平衡状態になっているとして計算を進めていることに由来する。

4. 終わりに

Haar et al. (1984) が作成したプログラムでは、音速 (ω)、第 2 virial 係数 (B)、等温条件下での Joule-Thomson 係数 (δ_T)、Joule-Thomson 係数 (μ) も計算できるようになっている。これらの計算は地球科学の研究ではあまり行われていないと考えたので、本報告で作成したプログラムには含めていない。いずれの計算も表 9 の関係式を利用すれば簡単に行うことができる。必要があれば、これらの計算を組み込めばよからう。

表 8 計算結果の出力例。出力値について矢印をつけて説明を加えている。

Units←入力した単位一覧

TEMPERATURE C

DENSITY g/cm³

PRESSURE bar

ENERGY J/g

Liquid phase←液相の性質

T= 300.0000 P= 85.837843 D= 0.712408946470

DP/DT=+0.10192948E+01 DP/DD=+0.437189577E+04

CP=+0.574555125E+01 CV=+0.30618178E+01

S= 3.253355 H= 1344.052607 U= 1332.003658

G= -520.607666 A= -532.656615

Vapor phase←気相の性質

T= 300.0000 P= 85.837843 D= 0.046153767054

DP/DT=+0.598053883E+01 DP/DD=+0.111106569E+04

CP=+0.598053883E+01 CV=+0.28512884E+01

S= 5.704188 H= 2748.747815 U= 2562.765504

G= -520.607656 A= -706.589967

出力中の各記号は以下の値を示す。

T: 入力した温度 (°C), P: 飽和蒸気圧 (bar), D: 密度 (g/cm³),

DP/DT: 密度一定の条件での圧力の温度微分 (bar/K),

DP/DD: 温度一定の条件下での圧力の密度微分 (bar•cm³/g),

CP: 定圧比熱 (J/g•K), CV: 定積比熱 (J/g•K),

S: エントロピー (J/g•K), H: エンタルピー (J/g), U: 内部エネルギー (J/g),

G: ギブスエネルギー (J/g), A: ヘルムホルツエネルギー (J/g)

表9 Haar et al. (1984) の式から計算できるその他の性質。

音速 (ω)

$$\omega = \left(\frac{c_p}{c_v} \frac{\partial P}{\partial \rho} \right)^{1/2}$$

第2 virial 係数(B)

$$B = \frac{1}{2RT} \left(\frac{\partial^2 P}{\partial \rho^2} \right)_{\rho=0}$$

等温条件下での Joule-Thomson 係数 (δ_T)

$$\delta_T \equiv \left(\frac{\partial h}{\partial P} \right)_T = \frac{1}{\rho} - \frac{T}{\rho^2} \left(\frac{\partial P / \partial T}{\partial P / \partial \rho} \right)$$

Joule-Thomson 係数 (μ)

$$\mu \equiv \left(\frac{\partial T}{\partial P} \right)_h = \frac{\delta_T}{c_p}$$

文献

- Haar, L., Gallagher, J. S., and Kell, G. S. (1984) NBS/NRC Steam Tables. 320p., Hemisphere Publishing, New York.
- 澁江靖弘 (1997) 高温・高圧条件におけるH₂OとH₂O+NaCl系熱水の密度式. 資源地質, **47**, 145-154.
- 澁江靖弘 (2003) H₂O+NaCl系およびH₂O+KCl系熱水溶液の飽和蒸気圧. 岩石鉱物科学, **32**, 185-191.
- 澁江靖弘 (2005) 気液二相共存条件下での水の熱力学的性質を計算するプログラム—Haar et al. (1984) の式を用いて—. 兵庫教育大学研究紀要, **26**, 105-117
- Wagner, W. and Pruß, A. (2002) The IAPWS formulation 1995 for the thermodynamic properties of ordinary water substance for general and scientific use. J. Phys. Chem. Ref. Data, **31**, 387-535.

List 1 プログラムリスト。

```

10000 REM HGK--2004.12.27
10050 DEFDBL A-H, M-Z
10100 DIM MG(40),II(40),JJ(40),BP(10),BQ(10)
10150 DIM ATZ(4),ADZ(4),AAT(4),AAD(4)
10200 DIM BV(10),A(8),C(18)
10250 DIM QR(11),QT(10),QZR(9),QZT(9)
10300 DIM FFD(2),FFP(5),FFH(5),NNT$(2),NND$(2),NNP$(5),NNH$(5)
10350 GOSUB *BLOCKDATA
10400 GOSUB *UNIT
10450 PRINT
10500 PRINT "Enter option and temperature. Option=1, input density. Option=2, input pressure."
10550 INPUT"Option No.";IOPT
10600 INPUT"Density or Pressure";X
10650 INPUT"Temperature";TT
10700 IF IOPT<>1 AND IOPT<>2 THEN PRINT:GOTO 10550
10750 IF X=0 THEN GOTO 16100
10800 T=TT
10850 GOSUB *TTTT
10900 T=TTT
10950 RT=GASCON*T
11000 GOSUB *BBT
11050 ON IOPT GOTO 11100, 11600
11100 DD=X
11150 D=DD*FD
11200 GOSUB *QQTD
11250 GOSUB *BASEDT
11300 ZBASE=BASEF
11350 DZ=DZB
11400 PRES=FP*(RT*D*ZBASE+Q)
11450 DQ=RT*(ZBASE+Y*DZ)+Q5
11500 Z=BASEF+Q/RT/D
11550 GOTO 12350
11600 PRES=X
11650 PINPUT=PRES/FP
11700 DGSS=P/T/.4#
11750 DLL=0
11800 DVV=0
11850 IF T>=TZ THEN DGSS=DLL:GOTO 12150
11900 DL=DLL:DV=DVV
11950 GOSUB *PCORRTPDLLDV
12000 IF T=<647.126# AND ABS((PINPUT-P)/P)=<5D-005 THEN PPP=P : GOTO 13500
12050 IF PINPUT>P THEN DGSS=DL:GOTO 12150
12100 IF PINPUT<P THEN DGSS=DV
12150 D=DGSS : PPP=PINPUT
12200 GOSUB *DFINDDOUTPDTDPD
12250 DD=DOUT/FD
12300 D=DD
12350 GOSUB *THERMDT
12400 U=UD*RT*FH
12450 H=HD*RT*FH
12500 S=SD*GASCON*FH*FT
12550 CP=CPD*GASCON*FH*FT
12600 CV=CVDX*GASCON*FH*FT
12650 DPDD=DQ*FD*FP
12700 DPDT1=DPDT*FP*FT
12750 G=GD*RT*FH
12800 A=AD*RT*FH
12850 LPRINT"Units"
12900 LPRINT A1$,SPC(3);NT$
12950 LPRINT A2$,SPC(3);ND$
13000 LPRINT A3$,SPC(3);NP$
13050 LPRINT A4$,SPC(3);NH$
13100 LPRINT
13150 LPRINT USING"T=#####.#### P=#####.##### D=#####";TT,PRES,DD

```

```

13200 LPRINT USING"DP/DT=+#.#####^" DP/DD=+#.#####^"; DPDT1, DPDD
13250 LPRINT USING"CP=+#.#####^" CV=+#.#####^"; CP,CV
13300 LPRINT USING"S=#####.##### H=#####.##### U=#####.#####";S, H, U
13350 LPRINT USING"G=#####.##### A=#####.#####";G, A
13400 LPRINT:LPRINT
13450 GOTO 16000
13500 D=DL
13550 IF T>646.3# THEN DD=D: GOTO 13700
13600 GOSUB *DFINDDOUTPDTDPD
13650 DD=DOUT/FD
13700 D=DD
13750 GOSUB *THERMDT
13800 U=UD*RT*FH
13850 H=HD*RT*FH
13900 S=SD*GASCON*FH*FT
13950 CP=CPD*GASCON*FH*FT
14000 CV=CVDX*GASCON*FH*FT
14050 DPDD=DQ*FD*FP
14100 DPDT1=DPDT*FP*FT
14150 PRES=PPP*FP
14200 G=GD*RT*FH
14250 A=AD*RT*FH
14300 LPRINT"Units"
14350 LPRINT A1$;SPC(3);NT$
14400 LPRINT A2$;SPC(3);ND$
14450 LPRINT A3$;SPC(3);NP$
14500 LPRINT A4$;SPC(3);NH$
14550 LPRINT
14600 LPRINT"Liquid phase"
14650 LPRINT USING"T=#####.#### P=#####.##### D=#####.#####";TT,PRES,DD
14700 LPRINT USING"DP/DT=+#.#####^" DP/DD=+#.#####^"; DPDT1, DPDD
14750 LPRINT USING"CP=+#.#####^" CV=+#.#####^"; CP,CV
14800 LPRINT USING"S=#####.##### H=#####.##### U=#####.#####";S,H,U
14850 LPRINT USING"G=#####.##### A=#####.#####";G, A
14900 D=DV
14950 IF T>646.3# THEN DD=D:GOTO 15100
15000 GOSUB *DFINDDOUTPDTDPD
15050 DD=DOUT/FD
15100 D=DD
15150 GOSUB *THERMDT
15200 U=UD*RT*FH
15250 H=HD*RT*FH
15300 S=SD*GASCON*FH*FT
15350 CP=CPD*GASCON*FH*FT
15400 CV=CVDX*GASCON*FH*FT
15450 DPDD=DQ*FD*FP
15500 DPDT1=DPDT*FP*FT
15550 G=GD*RT*FH
15600 A=AD*RT*FH
15650 LPRINT"Vapor phase"
15700 LPRINT USING"T=#####.#### P=#####.##### D=#####.#####";TT,PRES,DD
15750 LPRINT USING"DP/DT=+#.#####^" DP/DD=+#.#####^"; DPDT1, DPDD
15800 LPRINT USING"CP=+#.#####^" CV=+#.#####^"; CP,CV
15850 LPRINT USING"S=#####.##### H=#####.##### U=#####.#####";S,H,U
15900 LPRINT USING"G=#####.##### A=#####.#####";G, A
15950 LPRINT:LPRINT
16000 INPUT"Will you continue the calculation? Input Y(or y) or N(or n)";CAL$
16050 IF CAL$="Y" OR CAL$="y" THEN GOTO 10450
16100 END
16150 サブルーティン(*DFINDDOUTPDTDPD)
17700 *CORRTPDL DVDELG
17750 IF T>646.3# THEN GOTO 18650
17800 DLIQ=DLL
17850 IF DLL=<0 THEN DLIQ=1.11#-.0004*T
17900 DLL=DLIQ:D=DLIQ
17950 GOSUB *DFINDDOUTPDTDPD

```

```

18000 D=DOUT:DL=DOUT
18050 GOSUB *THERMDT
18100 GL=GD
18150 DVAP=DVV
18200 IF DVV=<0 THEN DVAP=P/RT
18250 D=DVAP:DVV=DVAP
18300 GOSUB *DFINDDOUTPDTDPD
18350 IF DOUT<5D-007 THEN DOUT=5D-007
18400 D=DOUT:DV=DOUT
18450 GOSUB *THERMDT
18500 GV=GD
18550 DELG=GL-GV
18600 RETURN
18650 P=0
18700 IF T>647.126# THEN RETURN
18750 DELG=0
18800 TAUC=.657128000000001*(1#-T/647.126#)^.325#
18850 DL=.322#+TAUC
18900 DV=.322#-TAUC
18950 D=DV
19000 GOSUB *BASEDT
19050 GOSUB *QQTD
19100 ZB=BASEF
19150 PPP=RT*D*DV*ZB+Q
19200 RETURN
19250 サブルーティン(*BBT)
20350 サブルーティン(*BASEDT)
21200 *QQTD
21250 QR(1)=0
21300 Q5=0
21350 Q=0
21400 AR=0
21450 DADT=0
21500 CVR=0
21550 DPDTR=0
21600 E=EXP((-1#)*AA*D)
21650 Q10=D*D*E
21700 Q20=1#-E
21750 QR(2)=Q10
21800 QV=TZ/T
21850 QT(1)=T/TZ
21900 FOR I=2 TO 10
21950 QR(I+1)=QR(I)*Q20
22000 QT(I)=QT(I-1)*QV
22050 NEXT I
22100 FOR I=1 TO INC
22150 K=II(I)+1
22200 L=JJ(I)
22250 QK=CDBL(K) : QL=CDBL(L)
22300 ZZQ=CDBL(K)
22350 QZR(K-1)=QR(K+1):QZT(L)=QT(L+1):QZR(K)=QR(K+2):QZT(L+1)=QT(L+2)
22400 QP=MG(I)*AA*QZR(K-1)*QZT(L)
22450 Q=Q+QP
22500 Q5=Q5+AA*(2#/D-AA*(1#-E*(QK-1#)/Q20))*QP
22550 AR=AR+MG(I)*QZR(K)*QZT(L)/Q10/ZZQ/RT
22600 DFDT=Q20^QK*(1#-QL)*QZT(L+1)/TZ/QK
22650 D2F=QL*DFDT
22700 DPT=DFDT*Q10*AA*QK/Q20
22750 DADT=DADT+MG(I)*DFDT
22800 DPDTR=DPDTR+MG(I)*DPT
22850 CVR=CVR+MG(I)*D2F/GASCON
22900 NEXT I
22950 QP=0
23000 Q2A=0
23050 FOR J=37 TO 40
23100 IF MG(J)=0 THEN GOTO 24350

```

```

23150 K=II(J)
23200 KM=JJ(J)
23250 QK=CDBL(K) : QKM=CDBL(KM)
23300 DDZ=ADZ(J-36)
23350 DEL=D/DDZ-1#
23400 IF ABS(DEL)<1D-010 THEN DEL=1D-010
23450 DDQQ=DEL*DEL
23500 EX1=(-1#)*AAD(J-36)*DEL^QK
23550 DEX=EXP(EX1)*DEL^QKM
23600 ATT=AAT(J-36)
23650 TX=ATZ(J-36)
23700 TAU=T/TX-1#
23750 EX2=(-1#)*ATT*TAU*TAU
23800 TEX=EXP(EX2)
23850 Q10=DEX*TEX
23900 QM=QKM/DEL-QK*AAD(J-36)*DEL^(QK-1#)
23950 FCT=QM*D*D*Q10/DDZ
24000 Q5T=FCT*(2#/D+QM/DDZ)-(D/DDZ)^2#*Q10*(QKM/DEL/DEL+QK*(QK-1#)*AAD(J-36)*DEL^(QK-2#))
24050 Q5=Q5+Q5T*MG(J)
24100 QP=QP+MG(J)*FCT
24150 DADT=DADT-2#*MG(J)*ATT*TAU*Q10/TX
24200 DPDTR=DPDTR-2#*MG(J)*ATT*TAU*FCT/TX
24250 Q2A=Q2A+T*MG(J)*(4#*ATT*EX2+2#*ATT)*Q10/TX/TX
24300 AR=AR+Q10*MG(J)/RT
24350 NEXT J
24400 SR=(-1#)*DADT/GASCON
24450 UR=AR+SR
24500 CVR=CVR+Q2A/GASCON
24550 Q=Q+QP
24600 RETURN
24650 サブルーティン(*THERMDT)
25450 サブルーティン(*PST)
26200 *IDEALT
26250 TIDEAL=T/100
26300 TL=LOG(TIDEAL)
26350 GI=(-1#)*(C(1)/TIDEAL+C(2))*TL
26400 HI=(C(2)+C(1)*(1#-TL)/TIDEAL)
26450 CPI=C(2)-C(1)/TIDEAL
26500 FOR I=3 TO 18
26550 GI=GI-C(I)*TIDEAL^CDBL(I-6)
26600 HI=HI+C(I)*CDBL((I-6))*TIDEAL^CDBL(I-6)
26650 CPI=CPI+C(I)*CDBL((I-6))*CDBL((I-5))*TIDEAL^CDBL(I-6)
26700 NEXT I
26750 AI=GI-1#
26800 UI=HI-1#
26850 CVIX=CPI-1#
26900 SI=UI-AI
26950 RETURN
27000 *PCORRTPDLVDV
27050 GOSUB *PST
27100 PPP=PS
27150 GOSUB *CORRTPDLVDVDELG
27200 DP=0
27250 DP=DELG*GASCON*T/(1#/DV-1#/DL)
27300 PPP=PPP+DP
27350 IF ABS(DELG)<.0001 THEN GOTO 27450
27400 DLL=DL:DVV=DV:GOTO 27150
27450 P=PPP
27500 RETURN
27550 サブルーティン(*UNIT)
28950 サブルーティン(*TTTT)
29350 サブルーティン(*BLOCKDATA)

```
